

Simple  friendly



川崎机器人控制器
E 系列

TCP/IP 通信功能
手册



川崎重工业株式会社

90210-1248DCA

前言

本手册介绍 E 系列控制器的 10BASE-T/100BASE-TX 以太网连接的 TCP/IP 通信功能。

本手册尽可能详细地介绍使用本功能的标准操作方法。但无论怎样，都不可能把所有需避免的可能操作、条件或情况都完全地介绍出来。所以，在操作中碰到任何未介绍的问题或情况时，请联络川崎公司。请参阅本手册封底上的联系信息。

在仔细阅读本手册的同时，还必须熟读与机器人一起发运的 E 控制器的基本手册（包括安全手册），只有当全部手册完整阅读并充分理解后，才能使用机器人。

-
1. 本手册并不构成对使用机器人的整个应用系统的担保。因此，川崎公司将不会对使用这样的系统而可能导致的事故、损害和(或)与工业产权相关的问题承担责任。
 2. 川崎公司郑重建议：所有参与机器人操作、示教、维护、维修、点检的人员，预先参加川崎公司准备的培训课程。
 3. 川崎公司保留未经预先通知而改变、修订或更新本手册的权利。
 4. 事先未经川崎公司书面许可，不可以将本手册全部或其中的一部分再版或复制。
 5. 请把本手册小心存放好，使之保持在随时备用状态。机器人如果需要重新安装、或搬运到不同地点、或卖给其他用户时，请务必将本手册附上。一旦出现丢失或严重损坏，请和您的川崎公司代理商联络。
-



Copyright © 2012 KAWASAKI HEAVY INDUSTRIES, LTD.

川崎重工 版权所有

符号

在本手册中，带下述符号的内容应特别注意。

为确保机器人的正确安全操作、防止人员伤害和财产损失，请遵守下述方框符号表达的安全信息。



目 录

前言	1
符号	2
1. 0 以太网连接的 TCP/IP 通信功能的概要	5
1. 1 远程终端功能	5
1. 2 套接字通信功能	5
1. 2. 1 TCP 通信命令	6
1. 2. 2 UDP 通信命令	7
1. 3 TCP/IP 通信支持规格	8
2. 0 以太网配置及与机器人控制器的连接	9
2. 1 无集线器直接连接	9
2. 2 用集线器连接网络	10
3. 0 IP 地址的设置与连接确认	11
3. 1 IP 地址和子网掩码的设置	11
3. 2 网络上的连接确认	13
4. 0 KCwinTCP/KRterm 远程终端功能的使用	15
4. 1 KCwinTCP/KRterm 是什么?	15
4. 2 系统配置要求	15
4. 3 软件的安装和机器人 IP 地址的登录	16
4. 3. 1 KCwinTCP 软件的安装和机器人 IP 地址的登录	16
4. 3. 2 KRTERM 的安装和机器人 IP 地址的登录	17
4. 4 KCwinTCP/Krterm 的操作	18
4. 4. 1 日常操作	18
4. 4. 2 菜单	20
4. 4. 2. 1 文件 (F)	20
4. 4. 2. 2 通信 (C)	21
4. 4. 2. 3 显示 (V)	24
4. 4. 2. 4 帮助 (H)	24
4. 4. 3 垂直/水平滚动条	24
4. 4. 4 标题显示	24
4. 5 注意及限制事项	25
4. 6 为了避免装载错误机器人数据	26
5. 0 DUP 的数据通信命令	28
5. 1 用 UDP_SENDTO 和 UDP_RECVFROM 命令产生的通信错误	33


6.0	TCP 中的数据通信命令	34
6.1	套接字通信中的通信错误.....	48
6.2	套接字通信命令使用例.....	50
6.2.1	当机器人为服务器时.....	50
6.2.2	当机器人为客户端时.....	53
7.0	错误信息.....	55

1.0 以太网连接的TCP/IP通信功能的概要

为了使用以下功能，用 10BASE-T/100BASE-TX 电缆将主CPU板的以太网端口与外围设备（PC等）连接从而来创建如下图所示的以太网。

远程终端功能

套接字通信功能



1.1 远程终端功能

通过在Windows计算机中安装由川崎公司提供的KCwinTCP/KRterm软件，除了RS232 通信（标准特征）的终端功能外，通过以太网连接的TCP/IP通信可以从PC输入AS监控指令等操作。

1.2 套接字通信功能

本通信功能提供基于TCP/IP通信用的套接字（socket）接口的命令。此功能能使数据在机器人控制器和其他设备间进行通信。

提供 2 种通信命令。

1. TCP通信命令
2. UDP通信命令


在TCP通信中，当通信错误产生时，自动地重新发送数据。当重视通信的确定性时使用该命令。在UDP通信中，错误产生后也不重新发送数据。当重视通信速度时使用该命令。

1.2.1 TCP 通信命令


有以下TCP通信命令：

- | | |
|------------------|--------------|
| • TCP_LISTEN | 创建套接字并等待连接请求 |
| • TCP_ACCEPT | 检查是否接收到连接请求 |
| • TCP_CONNECT | 创建套接字并发送连接请求 |
| • TCP_SEND | 发送数据字符串 |
| • TCP_RECV | 接收数据字符串 |
| • TCP_CLOSE | 中断套接字通信 |
| • TCP_END_LISTEN | 结束等待连接请求 |

例：机器人作为服务器




例：机器人作为客户



1.2.2 UDP 通信命令

有 2 种 UDP 通信命令：UDP_SENDTO（发送数据）和 UDP_RECVFROM（接收数据）。



1.3 TCP/IP 通信支持规格

1. 10BASE-T/100BASE-TX以太网
2. 网络连接传送速率: 10 Mbps/100 Mbps
3. 用RJ045 连接器连接
4. 协议: IP, ARP, RARP, ICMP, TCP, UDP, TELNET (服务器)

2.0 以太网配置及与机器人控制器的连接


本章以个人计算机（外部计算机）与机器人控制器连接为例介绍如何创建以太网。

可以用 2 种方法将个人计算机与机器人控制器相连接：

1. 不使用集线器将两台控制器连接到一台计算机上
2. 通过以太网用集线器将两台以上的计算机系统连接（包括机器人控制器）

2.1 无集线器直接连接


下图显示无集线器的连接。



在此情况下，使用 10BASE-T/100BASE-TX 交叉电缆连接。

2. 2 用集线器连接网络

下图显示用集线器连接 10BASE-T/100BASE-TX 电缆的网络。



在此情况下，用 10BASE-T/100BASE-TX 直式电缆连接。

3.0 IP地址的设置与连接确认

为了设置IP地址和确认连接，将机器人控制器和基于Windows操作系统的个人计算机与网络连接。连接硬件后，为了使用TCP/IP通信功能，一定要在网络上设置机器人控制器的IP地址。

3.1 IP 地址和子网掩码的设置

使用TCP/IP通信功能前，要登录控制器的IP地址和子网掩码。

为了设置IP地址和子网掩码，从示教器或RS232C连接的个人计算机的KCwin/KCwinTCP/KRterm远程终端画面来执行NETCONF监控指令。

下例为端口 1 的设置例。

例

```
>NETCONF [ ]
IP address = 192. 168. 0. 2
Change? (If not, press RETURN only)
172. 23. 123. 123 [ ]

IP address = 172. 23. 123. 123
Change? (If not, press RETURN only) [ ]

Host name =
Change? (If not, press RETURN only)
KAWASAKI [ ]

Host name = KAWASAKI
Change? (If not, press RETURN only) [ ]

Subnet mask = 255. 255. 240. 0
Change? (If not, press RETURN only)
255. 255. 255. 0 [ ]

Subnet mask= 255. 255. 255. 0
Change? (If not, press RETURN only) [ ]
Gateway IP = 0. 0. 0. 0
Change? (If not, press RETURN only)
```

255. 255. 255. 0

Gateway IP = 255. 255. 255. 0

Change? (If not, press RETURN only)

DNS server 1 IP = 0. 0. 0. 0

Change? (If not, press RETURN only)

172. 23. 123. 1

DNS server 1 IP = 172. 23. 123. 1

Change? (If not, press RETURN only)

DNS server 2 IP = 0. 0. 0. 0

Change? (If not, press RETURN only)

172. 23. 123. 2

DNS server 2 IP = 172. 23. 123. 2

Change? (If not, press RETURN only)

Domain name =

Change? (If not, press RETURN only)

kawasaki_robot

Domain name = kawasaki_robot

Change? (If not, press RETURN only)

MAC address = 00:09:0f:00:ff:01

(W1018) Turn control power ON/OFF to change network setting.

>

要设置端口 2，输入NETCONF 2：来代替NETCONF。

可以用示教器设置IP地址。请参阅《操作手册》(辅助功能 0812 网络设置)。

登录IP地址后，关闭控制器电源再打开来重新启动机器人控制器。

3.2 网路上的连接确认

一旦机器人控制器的以太网连接接口打开并正常启动，连接Windows 个人计算机到相同的网络，在MS-DOS对话框画面用ping 指令确认连接。在用ping 指令检查前，Windows 个人计算机需要设置以太网连接（登录IP地址和子网掩码）并重新启动。

1. Windows 个人计算机中打开MS-DOS Prompt画面。
2. 在MS-DOS对话框画面输入：

```
>ping IP_address
```

IP_address是要连接的机器人控制器的IP地址。如果正确连接的话，则显示以下信息。

例

```
C:>ping 172.023.003.004
Pinging 172.023.003.004 with 32 bytes of data:
Reply from 172.023.003.004: bytes=32 time=53ms TTL=64
Reply from 172.023.003.004: bytes=32 time=5ms TTL=64
Reply from 172.023.003.004: bytes=32 time=24ms TTL=64
Reply from 172.023.003.004: bytes=32 time=11ms TTL=64
Ping statistics for 172.023.003.004:
Packets: Sent = 4, Received = 4, Lost = 0 (0% loss),
Approximate round trip times in milli-seconds:
Minimum = 5ms, Maximum = 53ms, Average = 23ms
```

如果连接不正确的话，则将显示错误信息如超时错误。

例

```
C:>ping 172.023.003.005
Pinging 172.023.003.005 with 32 bytes of data:
Request timed out.
Request timed out.
Request timed out.
Request timed out.
Ping statistics for 172.023.003.005:
Packets: Sent = 4, Received = 0, Lost = 4 (100% loss),
Approximate round trip times in milli-seconds:
Minimum = 0ms, Maximum = 0ms, Average = 0ms
```

此错误因如下原因而产生。检查并修改错误，然后重试直到连接成功为止。

1. 集线器的电源未打开。
2. 网络电缆连接器故障或连接不正确。
3. 使用不适当的电缆。
4. 机器人或个人计算机的IP地址等设置不正确。

通常，集线器或以太网连接接口设备上有发光管（LED）。通过检查这些发光管的亮灯状况来检查信号状态。当不能连接时，使用这些故障排除信息。

4. 0 KCwinTCP/KRTERM远程终端功能的使用

KCwinTCP/KRterm有以太网连接的远程终端功能。当安装在Windows个人计算机上时，软件允许从个人计算机向机器人进行远程操作。在进行终端操作前，有必要用IP地址来确定哪一台机器人要使用。当机器人连接成功时，可以进行输入监控指令等的终端操作。在必要的操作完成后，用指定命令结束连接。

在1台控制器上一次可以连接3台个人计算机。换句话说，当连接1台控制器到3台个人计算机时，到终止连接为止不可能从其他个人计算机进行远程终端操作。.

KCwinTCP/KRterm远程终端功能的使用方法介绍如下。

4. 1 KCwinTCP/KRTERM 是什么？

KCwinTCP/KRterm是川崎机器人控制器专用的个人计算机终端软件。当在个人计算机上安装此软件并且用以太网连接个人计算机到控制器上时，通过TCP/IP通信可以进行以下操作。

- 从个人计算机输入AS监控指令
- 保存/装载控制器存储器内容到/从个人计算机等

4. 2 系统配置要求


硬件	搭载80486以上CPU（处理器），Microsoft Windows操作系统的个人计算机。
操作系统	KCwinTCP) Microsoft Windows 95/98/NT4.0/2000/XP KRterm) Microsoft Windows 2000/XP/Vista/7
已确认的可操作机型*	KCwinTCP) Toshiba Personal Computer Dynabook Satellite 2520 (Windows98) Compaq Armada 1500C (Windows98) IBM ThinkPad 365X (Windows95) KRterm) DELL Inspiron 8000 (Windows2000) Lenovo ThinkCentre (WindowXP) Gateway GT5228J (Windows Vista/7)

注* 对使用未经确认的机型川崎公司将不负任何责任。


4.3 软件的安装和机器人 IP 地址的登录

4.3.1 KCwinTCP 软件的安装和机器人 IP 地址的登录

在您的Windows个人计算机的理想路径中安装由川崎公司提供的(复制) KCwinTCP。安装后，双击KCwinTCP图标。KCwinTCP启动，如下窗口显示。



登录要连接的机器人控制器。从菜单栏中选择[Com (通信)] → [Options (选项)]。



在显示的窗口中，输入网络上的IP地址和机器人控制器名并点击<OK>。(可以登录几个。)

4.3.2 KRTERM的安装和机器人IP地址的登录


在您的Windows个人计算机的理想的路径中安装由川崎公司提供的(复制)KRterm软件(setupJ.exe)。之后,按照安装程序的说明进行安装。



安装完毕后,用你的鼠标选择KRterm图标并双击。KRterm启动,如下窗口显示。



登录要连接的机器人控制器。从菜单栏中选择[Com (通信)]→[Options (选项)]。在显示的窗口中,输入网络上的IP地址和机器人控制器名并点击<OK>。(可以登录几个。)




4.4 KCwinTCP/KRTERM 的操作

4.4.1 日常操作

1. 打开控制器电源，不要初始化控制器。TELNET服务器功能启动并且控制器正在等待从客户个人计算机来的连接。


2. 启动KCwinTCP/KRterm

双击KCwinTCP/KRterm 图标。KCwinTCP/KRterm启动, 如下窗口显示。在此窗口的菜单栏中选择[Com (通信)] → [Sel Connection (选择连接) (E)]。



用光标键选择要连接的机器人的IP地址。点击 <OK>。

个人计算机试图连接指定的IP地址。当连接成功时，注册：显示以下几个信息。在此信息后输入“as”，并且机器人将返回对话框。



现在，可以从个人计算机输入AS监控指令。

当在个人计算机和控制器间上载或下载示教数据时，输入与AS终端中相同的SAVE/LOAD（保存/装载）监控指令。


- 当执行 SAVE（保存）指令时

```
>
>
>
>save data01.as
存盘中…(data01.as)
网络配置数据
网络配置数据
机器人数值
运行信息数据
系统数值
辅助数值
人机界面数值
接口面板页标题
接口面板用颜色数据
信号名称
存储数值
程序 pg1
程序 pg2
文件存盘完毕
|
```

- 当执行 LOAD（装载）指令时

```
>
>load data01.as
装载中…(data01.as)
机器人数值 装载
确认吗?(是:1, 否:0)1
机器人数值
装载中…
运行信息数据
系统数值
辅助数值
人机界面数值
接口面板页标题
接口面板用颜色数据
信号名称
存储数值
程序 pg1()
程序 pg2()
(E1048) 调零偏差数据非法。
文件装载完毕。(0 errors)
>
```


在所有操作结束终止。从菜单栏中选择[Communication (通信)] → [Disconnect (断开)]。



4.4.2 菜单

4.4.2.1 文件 (F)

Open log file (打开日志文件) (O)



为了保存作为日志文件在画面上显示的字符, 指定上述中的一个文件。(日志文件的默认扩展名为.log.)

Close log file (关闭日志文件) (C)
结束日志。


Exit (退出) (X)
结束KCwinTCP/KRterm。

4.4.2.2 通信 (C)

Connect (连接) (C)

与上一次使用了IP地址和端口编码的机器人建立连接。

Sel(ect) Connect(ion) (选择连接) (E)



在一览表中显示在[Options (选项) (O)]或[Sel. Connect (选择连接)]中登录的IP地址。选择指定的IP地址并点击〈OK〉来与那台计算机建立连接。

〈OK〉

与选择的IP地址连接。

〈Cancel (取消) 〉


不进行任何连接而结束。

Disconnect (断开) (D)

中断连接。

Options (选项) (0) (仅KCwinTCP显示)

登录网络上的IP地址和机器人名。最多能登录256台机器人。



IP address (IP 地址)

输入要连接的机器人的IP地址。

Name (名称)

输入要连接的机器人名(昵称)。

<Add (添加) >

在一览表中添加输入的IP地址和名称。

<Insert (插入) >

在光标行插入IP地址和名称。

<Edit (编辑) >

用新的内容更新选择的IP地址和名称。

<Delete (删除) >

从一览表中删除选择的IP地址和名称。

<OK>


更新内容并保存所有更改的内容到 KCWINTCP. INI 后，结束选项画面。

<Cancel (取消) >

不保存更改的内容而结束画面。(KCWINTCP. INI 不更新。)

Options (选项) (0) (仅KRterm显示)

登录网络上的IP地址和机器人名。最多能登录256台机器人。



IP address (IP地址)

输入要连接的机器人的IP地址。

Port number (端口号)

默认输入端口号23 (来连接TELNET)。

Name (名称)

输入要连接的机器人名(昵称)。

<Add (添加) >

在一览表中添加输入的IP地址和名称。

<Insert (插入) >

在光标行插入IP地址和名称。

<Edit (编辑) >

用新的内容更新选择的IP地址和名称。

<Delete (删除) >

从一览表中删除选择的IP地址和名称。

<OK>

更新内容并保存所有更改的内容到KRterm. INI后，结束选项画面。

<Cancel (取消)>
不保存更改的内容而结束画面。(KRterm. INI不更新。)

4.4.2.3 显示 (V)

Font (字体) (F)
选择在画面上使用的字符的字体。

Screen (画面) (W)
设置滚动缓冲区数和画面背景的颜色和文本颜色。

4.4.2.4 帮助 (H)

Contents (目录) (C)
打开KCwinTCP/KRterm操作说明的目录。

About Kcwin (有关KCwin) (A)
显示KCwinTCP/KRterm版本信息。

4.4.3 垂直/水平滚动条

当数据超过画面行数时，显示垂直滚动条。使用滚动条KCwinTCP可以最多显示200行，KRterm可以最多显示1000行。

4.4.4 标题显示

在标题栏中显示以下信息。

例



[192. 168. 29. 128:econ]

显示当前客户在格式[客户端IP地址：名称]。当未建立连接时，在此不显示任何内容。

[D:\KRTerm]

如果在AS语言LOAD/SAVE 指令中仅指定文件名的话, 到/从此路径中保存/装载数据。通过菜单→[File (文件) (F)]→[Set current folder (当前文件夹的设定) (S)]来更改路

径。

[Logging (记录中)]

当保存画面数据到日志文件中时显示。

4.5 注意及限制事项

在监控指令执行的过程中终止连接时，机器人控制器将用 ABORT 指令终止该指令(如：EDIT)。

到执行指令为止，不要从个人计算机断开。如果因误操作而断开的话，请立即重新连接并检查是否正确地执行该指令。

当从机器人终止连接时(如：OS 显示断开信息)，KCwinTCP/KRterm 自动地断开连接。如果其不自动地断开，则手动断开 KCwinTCP/KRterm。

KCwinTCP/KRterm 不支持 AS 指令 FDIRECTORY, FDELETE, FORMAT (用 Windows Explorer 进行操作。)

4.6 为了避免装载错误的机器人口数据

因连接多台机器人到网络上，将有机会装载错误的机器人口数据到机器人中。因此，错误的机器人口数据有可能更新已保存的文件。

为了避免这些，KCwinTCP/KRterm 检查装载前的数据是否正确。当保存文件时，在数据的开始保存 IP 地址数据（参阅下面以.NETCONF 开始的行）。

```
*****
.*== AS 组 ==      : ASE_021000VOP 2011/12/17 21:40
.*用户 IF AS        : UASE021000VOP 2011/12/17 21:40
.*用户 IF TP        : UTPE021000VOP 2011/12/17 21:40
.*手臂控制 AS      : AASE021000VOP 2011/12/17 21:40
.*用户 IF AS 信息文件  : MASE0200VOPCH 2011/12/17 21:39
.*用户 IF TP 信息文件  : MTPE0200VOPCH 2011/12/17 21:39
.*手臂数据文件      : ARME021000VOP 2011/12/17 21:42
.*内核_KNL101600000 2011/07/26
.*驱动程序_DRV102700000 2011/09/15
.*RFS                : _RFS100700100 2011/08/01
.*== 伺服组 ==      : SVE_08000001Q 2011/12/16 21:19
.*手臂控制 伺服      : ASVE08000001Q 2011/12/16 21:13
.*伺服数据文件      : ASPE08000001Q 2011/12/16 21:13
.*手臂控制伺服 FPGA: ASFE080000005 2008/12/08 13:28
.*
.*Cpu board type     : 1TA
.*== I/F board No. 1 == :
.*  [出厂设定数据]
.*没有出厂设定数据。
*****
.NETCONF    157.115.242.234, "KAWASAKI", 255.255.255.0, 192.168.0.1, 0.0.0.0, 0.0.0.0, " "
.NETCONF2   192.168.11.2, 255.255.255.0, 192.168.11.1
.ROBOTDATA1
ZROBOT.TYPE 30 30 6 1      -1006 RS030N-A001 ( 2011-12-29 13:07 )
ZSYSTEM      1 5 1          -106
ZLINEAR      0 0 0 0 0 -1 -1 -1 -1 -1 -1 -1 -1 -1 -1 -1
179
CONT_CODE    701      -10516
ZZERO        268435456 268435456 268435456 268435456 268435456 268435456 268435456 268435456
268435456 268435456 268435456 268435456 268435456 268435456 268435456 268435456 268435456
268435456 536870911
```

KCwinTCP/KRterm 检查以.NETCONF 开始的 IP 地址是否与控制器的 IP 地址相同。如果其不同的话，KCwinTCP/KRterm 询问是否将继续装载。检查如果数据正确的话，则输入 9 继续，输入 0 结束。



5.0 DUP的数据通信命令

以下命令通过以太网 UDP/IP 在机器人控制器和计算机系统间进行数据通信。

UDP_SENTO 发送数据的程序命令

以 UDP/IP 协议为基准发送指定的字符串数据。此命令创建套接字，发送数据并一连串关闭套接字。

UDP_RECVFROM 接收数据的程序命令

以 UDP/IP 协议为基准接收并保存数据到指定的字符串数组中。此命令创建套接字，接收数据并一连串关闭套接字。

示例

关键字

参数

UDP_SENTO 返回值变量, IP 地址数组变量, ...

省略输入用  标记的参数值。

必须在关键字和第一个参数之间输入一个空格。

示例中  代表回车键(Enter 键)。

程序命令

UDP_SENTO 返回值变量, IP 地址数组变量, 端口号,
发送数据的字符串变量数组, 元素数, 超时时间

功能

以 UDP 协议为基准发送数据字符串。用字符串变量数组指定要发送的数据。此命令一连串创建套接字，发送数据并关闭套接字。如果通信错误产生，在返回值存储变量中保存错误代码并且程序执行不停止。

参数

返回值变量

- 指定保存执行结果的变量。
- 当执行正常结束时，保存为 0。
- 当通信错误产生时，保存错误代码。但，程序继续执行。
- 当无通信错误产生时，不保存错误代码（如：参数错误）。在这种情况下，程序执行导致出错并停止。
- 有关通信错误代码的详情，请参阅 5.1。

IP地址数组变量

- 指定保存目标计算机系统的IP地址的数组变量的第一元素。以 8 位增值保存IP地址到数组变量的各元素。
- 超过 0-255 的数据视为错误。如果无错误产生，第一数据必须在 0 - 247 间。

端口号

- 指定表示连接对方（进程）的通道的端口号。用该端口号打开套接字。
- 容许范围为 8192 - 65535，否则错误产生。

发送数据的字符串变量数组

- 指定保存要发送数据的字符串变量数组的第一元素。
- 按先后顺序发送数组元素。
- 将要发送的数据更改到字符串格式。可以发送转换为字符串格式的数值数据。确保编码格式与接收方相一致（大端/小端，等）。
- 1 个命令可以发送的最多数据量为 1472 字节。

元素数

- 指定保存要发送的字符串变量数组的元素数。
- 以索引号的顺序发送这些数组元素给接收方。

超时时间

- 当未连接时, 指定输出超时错误前, 要等待的时间(秒)。默认值为 1 秒。
- 当超时错误产生时, 将错误代码保存到返回值存储变量。
- 即使超时错误产生, 也不停止AS语言的执行。

示例

每当执行机器人操作时, 增加计数。

```
.PROGRAM pg1
:
运动命令
:
count = count + 1
$cnt[0] = $ENCODE(/D,count)
CALL send(.[$cnt[]])
.END
```

发送机器人执行的运动次数。

```
.PROGRAM send(.[$cnt[]])
UDP_SENDTO ret,ip[0],49152,.[$cnt[0]],1,2
IF ret<>0 THEN
    TYPE "ERROR!"
    HALT
END
.END
```

程序命令

UDP_RECVFROM 返回值变量, 端口号, 接收数据的字符串变量数组,
元素数, 超时时间, IP地址数组变量, 最大字节数

功能

以 UDP 协议为基准接收并保存数据到字符串变量数组中。此命令一连串创建套接字，接收数据并关闭套接字。如果通信错误产生，在返回值存储变量中保存错误代码并且程序执行不停止。

参数

返回值变量

- 指定保存执行结果的变量。
- 当执行正常结束时，保存为 0。
- 当通信错误产生时，保存错误代码。但，程序继续执行。
- 当无通信错误产生时，不保存错误代码（如：参数错误）。在这种情况下，程序的执行将导致出错并停止。
- 有关通信错误代码的详情，请参阅 5.1。

端口号

- 指定表示连接对方（进程）的通道的端口号。用该端口号打开套接字。
- 容许范围为 8192 – 65535，否则错误产生。

接收数据的字符串变量数组

- 指定保存已接收数据的字符串变量数组的第一元素。
- 接收的数据为每一字节的连续的字符串数据。当接收转换为字符串格式的数值数据时，确保编码格式与发送方相一致（大端/小端，数据分割等）。
- 1 个命令可以发送的最多数据量为 1472 字节。

元素数

- 设置变量来代替接收的字符串变量数组的元素数。
- 指定为变量。

超时时间

- 当未连接时，指定输出超时错误前要等待的时间(秒)。默认值为 1 秒。
- 当超时错误产生时，将错误代码保存到返回值存储变量。
- 即使超时错误产生，也不停止AS语言的执行。

IP地址数组变量

- 指定数组变量的第一元素来保存发送数据的节点的IP地址的。将IP地址的 32 位分为各为 8 位的 4 个元素，并且保存各元素到指定数组的数组元素中。

最大字节数

- 指定保存在字符串变量数组元素的最大字节数。
- 容许范围为 1-255，否则错误产生。默认值为 255 字节。

示例

```
由.start_flg 的值机器人运动。
.PROGRAM pg1
CALL rcv(.start_flg)
IF .start_flg == 1 THEN
:
运动命令
:
END
.END

以从其他机器人接收的操作为基准操作机器人。
.PROGRAM rcv(.start_flg)
UDP_RECVFROM ret, 49152, $cnt[0], p, 1, ip[0], 255
IF ret<>0 THEN
    TYPE "ERROR!"
    HALT
END
num = VAL($cnt[0])
IF num%10 == 0 THEN
    .start_flg = 1
ELSE
    .start_flg = 0
END
.END
```

5.1 用 UDP_SENDTO 和 UDP_RECVFROM 命令产生的通信错误

代码	错误信息	说明	对策
E4019	TCPIP) 插口 (套接字) 打开错误。代码=XX Dst. IP=XX. XX. XX. XX	当执行 UDP_SENDTO/UDP_RECVFROM 命令时, 未能打开套接字。已打开太多的套接字。或, 参数错误如错误的 IP 地址。	此命令不要执行的太多。 修改错误的参数。 CODE (代码) 表示 TCP/IP 包的 SUB 错误代码。
E4020	TCPIP) 插口 (套接字) 关闭错误。代码=XX Dst. IP=XX. XX. XX. XX	当执行 UDP_SENDTO/UDP_RECVFROM 命令时, 未能关闭套接字。可能是以太网上的问题或连接对方的问题。	检查连接对方的节点或连接线路是否有问题, 如果需要的话, 则进行修改。 CODE (代码) 表示 TCP/IP 包的 SUB 错误代码。
E4021	TCPIP) 通信错误。代 码=XX Dst. IP=XX. XX. XX. XX	在执行 UDP_SENDTO/UDP_RECVFROM 命令中, 通信错误产生。可能是以太网上的问题或连接对方的问题。	检查连接对方的节点或连接线路是否有问题, 如果需要的话, 则进行修改。 CODE (代码) 表示 TCP/IP 包的 SUB 错误代码。
E4022	TCPIP) 消息 (信息) 太长。	当执行 UDP_SENDTO/UDP_RECVFROM 命令时, 试图发送的数据超过极限。	修改数据以便其数据量在极限范围内并重试程序。
E4023	TCPIP) 不能到达主机 (上位机)。	当执行 UDP_SENDTO/UDP_RECVFROM 命令时, 不能与上位机连接。在网络上无该 IP 地址的上位机。端口号或 IP 地址可能错误。可能是以太网上的问题或连接对方的问题。	检查端口号或 IP 地址是否正确, 如果必要的话, 则进行修改。检查连接对方的节点或连接线路是否有问题, 如果需要的话, 则进行修改。
E4024	TCPIP) 通信时间超 时。 Dst. IP=XX. XX. XX. XX	在执行 UDP_SENDTO/UDP_RECVFROM 命令的过程中, 超时产生。可能是以太网上的问题或连接对方的问题。	检查连接对方的节点或连接线路是否有问题, 如果需要的话, 则进行修改。
E4025	TCPIP) 连接失败。	当执行 UDP_SENDTO/UDP_RECVFROM 命令时, 中断连接。可能是以太网上的问题或连接对方的问题。	检查连接对方的节点或连接线路是否有问题, 如果需要的话, 则进行修改。
E4026	TCPIP) 无缓冲空间。	在执行 UDP_SENDTO/ UDP_RECVFROM 命令的过程中, 通信错误产生。TCP/IP 系统资源 (队列, 缓冲器) 缺乏。或, 可能是以太网上的问题或连接对方的问题。	同时, 停止正在运行中的一些 TCP/IP 应用。重新启动控制器。
E4027	TCPIP) 坏的插口。	当执行 UDP_SENDTO/UDP_RECVFROM 命令时, 不能连接。端口号或 IP 地址可能错误。或, 可能是以太网上的问题或连接对方的问题。	检查端口号或 IP 地址是否正确, 如果必要的话, 则进行修改。检查连接对方的节点或连接线路是否有问题, 如果需要的话, 则进行修改。

6.0 TCP 中的数据通信命令

有以下命令通过以太网 TCP 在机器人控制器和其他计算机系统间允许数据通信。

TCP_LISTEN 开始等待连接请求的程序命令（服务器侧开始通信服务时使用）
指定端口号并创建套接字，该套接字等待连接请求。当机器人控制器作为服务器使用时使用。

用以下 TCP_ACCEPT 命令检查是否接收到连接请求。

TCP_ACCEPT 检查连接请求的程序命令（服务器侧开始通信服务时使用）
检查指定端口是否接收到套接字通信的连接请求。当指定端口接收到连接请求时，此命令结束。

用 TCP_LISTEN 命令创建套接字后，使用 TCP_ACCEPT 命令。用此命令确认连接后，使用以下的 TCP_SEND 和 TCP_RECV 命令来发送/接收数据。

TCP_CONNECT 要求连接的程序命令（客户端开始通信服务时使用）
指定端口号并创建套接字。然后，发送套接字通信的连接请求到指定节点，建立连接。
当机器人是客户时使用 TCP_CONNECT。

用 TCP_CONNECT 命令建立连接后，再用 TCP_SEND 和 TCP_RECV 命令来发送/接收数据到/从服务器。

TCP_SEND 发送数据的程序命令
以 TCP 协议为基准发送指定的字符串数据。

用 TCP_ACCEPT 或 TCP_CONNECT 命令连接后，使用 TCP_SEND 命令。

TCP_RECV 接收数据的程序命令
以 TCP 协议为基准接收并保存数据到指定的字符串变量中。

用 TCP_ACCEPT 或 TCP_CONNECT 命令连接后，使用 TCP_RECV 命令。

TCP_CLOSE 终止连接的程序命令（结束通信服务时使用）
关闭当前的套接字连接。用以下的 TCP_END_LISTEN 命令关闭用 TCP_LISTEN 命令创建的套接字。

TCP_END_LISTEN 终止连接的程序命令

关闭当前等待连接请求的套接字。用上述的 TCP_CLOSE 命令关闭已连接的套接字。

示 例

关键字

参数



TCP_ACCEPT 返回值变量, 端口号, 超时时间, ...

可以省略输入用 标记的参数值。

必须在关键字和第一个参数之间输入一个空格。

示例中代表回车键(Enter 键)。

程序命令

TCP_LISTEN 返回值变量, 端口号

功能

开始等待连接请求的程序命令（服务器侧开始通信服务时使用）。

指定端口号并创建套接字，该套接字等待连接请求。在执行过程中如果通信错误产生，在返回值保存变量中保存错误代码并且程序执行不停止。

参数

返回值变量

- 指定保存执行结果的变量。
- 当执行正常结束时，保存为 0。
- 当通信错误产生时，保存错误代码。但，程序继续执行。
- 当无通信错误产生时，不保存错误代码（如：参数错误）。在这种情况下，程序的执行将导致出错并停止。
- 有关通信错误代码的详情，请参阅 6.1。

端口号

- 指定表示连接对方的通道的端口号。用该端口号打开套接字。
- 容许范围为 8192 – 65535，否则错误产生。

程序命令

TCP_ ACCEPT 返回值变量, 端口号, 超时时间, 客户端 IP 地址数组变量

功能

检查连接请求的程序命令（服务器侧开始通信服务时使用）

检查指定端口是否接收到套接字通信的连接请求，如果已接收到，则建立连接。当此命令正常结束时，连接已建立。在执行过程中如果通信错误产生，在返回值保存变量中保存错误代码并且程序执行不停止。

参数

返回值变量

- 指定保存执行结果的变量。
- 当执行正常结束时，保存为 0。
- 当通信错误产生时，保存错误代码。但，程序继续执行。
- 当无通信错误产生时，不保存错误代码（如：参数错误）。在这种情况下，程序的执行将导致出错并停止。
- 有关通信错误代码的详情，请参阅 6.1

端口号

- 指定表示连接对方（进程）的通道的端口号。用该端口号打开套接字。
- 容许范围为 8192–65535，否则错误产生。

超时时间

- 当未建立连接时，指定输出超时错误前要等待的时间(秒)。默认值为 1 秒。容许范围为 0–60，否则错误产生。如果未指定，则为 1 秒。
- 当超时错误产生时，将错误代码保存到返回值存储变量。
- 即使超时错误产生，也不停止AS语言的执行。

客户端IP地址数组变量

- 当连接成功时，指定保存客户端（32 位）的IP地址的数组变量。
- 按从IP地址开始的顺序以 8 位增值将IP地址保存到 数组变量的各元素。

示例

当指定数组变量为 ipa[0]时，并且与计算机“192.168.0.3”建立连接时，保存IP地址到数组变量元素ipa[0]至ipa[3]。

```
ipa[0] : 192
ipa[1] : 168
ipa[2] : 0
ipa[3] : 3
```

程序命令

TCP_CONNECT 返回值变量, 端口号, 服务器 IP 地址数组变量, 超时时间

功能

要求连接的程序命令（客户端开始通信服务时使用）

指定端口号并创建套接字。然后，发送套接字通信的连接请求到指定节点，建立连接。
由服务器的 IP 地址决定节点。

在执行过程中如果通信错误产生，在返回值保存变量中保存错误代码并且程序执行不停止。

参数

返回值变量

- 指定保存执行结果的变量。
- 当执行正常结束时，保存为套接字号(socket ID)（0 以上的数字）。当通信错误产生时，保存错误代码。但，程序继续执行。
- 当无通信错误产生时，不保存错误代码（如：参数错误）。在这种情况下，程序的执行将导致出错并停止。
- 有关通信错误代码的详情，请参阅 6.1。

端口号

- 指定表示连接对方（进程）的通道的端口号。用该端口号打开套接字。
- 容许范围为 8192–65535，否则错误产生。

服务器IP地址数组变量

- 当连接成功时，指定保存服务器（32 位）的IP地址的数组变量。
- 按从IP地址开始的顺序以 8 位增值将IP地址保存到数组变量的各元素。

超时时间

- 当未建立连接时，指定输出超时错误前要等待的时间(秒)。默认值为 1 秒。容许范围为 0–60，否则错误产生。如果未指定，则为 1 秒。
- 当超时错误产生时，将错误代码保存到返回值存储变量。
- 即使超时错误产生，也不停止AS语言的执行。

示例

当指定数组变量为 ipa[0]时，并且与计算机“192.168.0.3”建立连接时，保存IP地址到数组变量元素ipa[0]至ipa[3]。

```
ipa[0] : 192  
ipa[1] : 168  
ipa[2] : 0  
ipa[3] : 3
```

程序命令

TCP_SEND 返回值变量, 套接字号, 发送数据的字符串变量数组, 元素数,
超时时间

功能

发送数据的程序命令

以TCP协议为基准发送数据。用字符串变量数组指定要发送的数据。在执行过程中如果通信错误产生，在返回值保存变量中保存错误代码并且程序执行不停止。

参数

返回值变量

- 指定保存执行结果的变量。
- 当执行正常结束时，保存为 0。
- 当通信错误产生时，保存错误代码。但，程序继续执行。
- 当无通信错误产生时，不保存错误代码（如：参数错误）。在这种情况下，程序的执行将导致出错并停止。
- 有关通信错误代码的详情，请参阅 6.1。

套接字号

- 指定用TCP_ACCEPT或TCP_CONNECT 命令取得的套接字号。

发送数据的字符串变量数组

- 指定保存要发送数据的字符串变量数组的第一元素。
- 按从头到尾的顺序发送数组元素。
- 将要发送的数据转换为字符串变量格式。可以发送转换为字符串格式的数值数据。确保编码格式与接收方相一致（大端/小端，等）。
- 1 个命令可以发送的最多数据量为 4096 字节。

元素数

- 指定要发送的字符串变量数组的元素数。
- 以索引号的顺序发送指定的元素数到接收方。

超时时间

- 当未建立连接时，指定输出超时错误前要等待的时间(秒)。默认值为 1 秒。
- 容许范围为 0-60，否则错误产生。如果未指定，则为 1 秒。
- 当超时错误产生时，将错误代码保存到返回值存储变量。
- 即使超时错误产生，也不停止AS语言的执行。

程序命令

**TCP_RECV 返回值变量, 套接字号, 接收数据存储用的字符串变量数组, 元素数,
超时时间, 最大字节数**

功能

接收数据的程序命令

以TCP协议为基准接收数据并保存该数据到指定的字符串变量数组中。在执行过程中如果通信错误产生，在返回值保存变量中保存错误代码并且程序执行不停止。

参数

返回值变量

- 指定保存执行结果的变量。
- 当执行正常结束时，保存为 0。
- 当通信错误产生时，保存错误代码。但，程序继续执行。
- 当无通信错误产生时，不保存错误代码（如：参数错误）。在这种情况下，程序的执行将导致出错并停止。
- 有关通信错误代码的详情，请参阅 6.1。

套接字号

- 指定用TCP_ACCEPT或TCP_CONNECT 命令取得的 套接字号。

接收数据存储用的字符串变量数组

- 指定保存要接收数据的字符串变量数组的第一元素。
- 接收的数据为每一字节的连续的字符串数据。当接收转换为字符串格式的数值数据时，确保编码格式与发送方相一致（大端/小端，数据分割等）。
- 1 个命令可以发送的最多数据量为 4096 字节。

数组元素个数

- 设置变量来代替接收的字符串变量数组的数组元素个数。
- 用变量指定。

超时时间

- 当未建立连接时，指定输出超时错误前要等待的时间(秒)。默认值为 1 秒。容许范围为 0–60，否则错误产生。如果未指定，则为 1 秒。
- 当超时错误产生时，将错误代码保存到返回值存储变量。
- 即使超时错误产生，也不停止AS语言的执行。

最大字节数

- 指定保存在字符串变量数组元素的最大字节的数。
- 容许范围为 1-255，否则错误产生。
- 默认值为 255 字节。

示例

当指定接收数据存储用字符串变量数组为 “\$recv_dt[0]”，数组元素个数为 “rcv_cnt”，及元素的最大字节数为 10 时，并且当接收到字符串 “abcdefghijklmнопqrstuvwxyz” 时，保存此数据到如下所示的接收数据存储用字符串变量数组的数组元素 \$recv_dt[0]至\$recv_dt[3]。

```
$recv_dt[0]: "abcdefghijklj"  
$recv_dt[1]: "klmnopqrst"  
$recv_dt[2]: "uvwxyz"  
rcv_cnt : 3
```

程序命令

TCP_CLOSE 返回值变量, 套接字号

功能

终止连接的程序命令（结束通信服务时使用）

切断套接字通信的连接并关闭套接字。在执行过程中如果通信错误产生，在返回值保存变量中保存错误代码并且程序执行不停止。

当因通信错误等停止TCP_CLOSE执行时，不可能完全断开连接。在这种情况下，再次执行TCP_CLOSE命令。其将内部关闭（强制关闭）连接。

参数

返回值变量

- 指定保存执行结果的变量。
- 当执行正常结束时，保存为 0。
- 当通信错误产生时，保存错误代码。但，程序继续执行。
- 当无通信错误产生时，不保存错误代码（如：参数错误）。在这种情况下，程序的执行将导致出错并停止。
- 有关通信错误代码的详情，请参阅 6.1。

套接字号

- 指定用TCP_ACCEPT或TCP_CONNECT 命令取得的套接字号。

程序命令

TCP_END_LISTEN 返回值变量，端口号

功能

终止连接的程序命令

结束用 TCP_LISTEN命令指定的套接字的 等待连接请求 并关闭套接字。在执行过程中如果通信错误产生，在返回值保存变量中保存错误代码并且程序执行不停止。

参数

返回值变量

- 指定保存执行结果的变量。
- 当执行正常结束时，保存为 0。
- 当通信错误产生时，保存错误代码。但，程序继续执行。
- 当无通信错误产生时，不保存错误代码（如：参数错误）。在这种情况下，程序的执行将导致出错并停止。
- 有关通信错误代码的详情，请参阅 6.1。

端口号

- 指定当前等待连接请求的套接字的端口号（用TCP_LISTEN命令指定的套接字）。

程序命令

**TCP_STATUS 返回值变量，端口号 数组 变量，套接字号数组 变量，
错误代码数组 变量，错误子代码数组 变量，IP地址数组 变量**

功能

将在TCP通信命令中使用的套接字状态保存到指定的数组变量。在机器人控制器内部控制的套接字号的顺序保存数据。在各参数中用相同数组号的数据是来自同一套接字的数据。当参数中有错误时，程序执行停止。

参数

返回值存储变量

- 保存使用的套接字数。
- 当参数中有错误时，不保存数据。

端口号数组变量

- 保存使用的套接字的端口号。
- 仅保存用TCP_LISTEN, TCP_CONNECT 命令创建的数据。保存用TCP_ACCEPT命令创建的套接字数据为 0。

套接字号数组变量

- 保存使用的套接字的套接字号。

错误代码数组变量

- 保存最近产生的通信错误的错误代码。
- 其包括未在 6.10 中介绍的内部控制器TCP/IP错误代码。

错误子代码数组变量

- 保存错误代码的详细代码。
- TCP/IP包的错误子代码。

IP地址数组变量

- 用字符串保存连接设备的IP地址。
例：192.168.0.1
- 仅保存用TCP_ACCEPT, TCP_CONNECT 命令创建的数据。保存用TCP_LISTEN命令创建的套接字数据为“0.0.0.0”。

示例

端口号	套接字号	错误代码	详细代码	IP 地址
8192	1	0	0	0; 用TCP_LISTEN命令创建的套接字
0	2	0	0	192.168.0.2; 用TCP_ACCEPT命令创建的套接字
49152	3	0	0	192.168.0.5; 用TCP_CONNECT命令创建的套接字
49153	4	-21670	90	192.168.0.4; 错误状态下的套接字的例

当以上述条件执行以下命令时，保存数据如下。

TCP_STATUS sock_cnt, port_no[0], sock_id[0], err_cd[0], sub_cd[0], \$ip_adrs[0]

```
sock_cnt      : 4
port_no[0]    : 8192
port_no[1]    : 0
port_no[2]    : 49152
port_no[3]    : 49153
sock_id[0]    : 1
sock_id[1]    : 2
sock_id[2]    : 3
sock_id[3]    : 4
err_cd[0]     : 0
err_cd[1]     : 0
err_cd[2]     : 0
err_cd[3]     : -21670
sub_cd[0]     : 0
sub_cd[1]     : 0
sub_cd[2]     : 0
sub_cd[3]     : 90
$ip_adrs[0]   : 0.0.0.0
$ip_adrs[1]   : 192.168.0.2
$ip_adrs[2]   : 192.168.0.5
$ip_adrs[3]   : 192.168.0.4
```

6.1 套接字通信中的通信错误

在套接字通信命令执行时，即使检测出以下错误，也不停止程序的执行。在错误日志中记录所有的错误。

注* 当控制器在执行TCP_ACCEPT命令时，如果错误产生的话，程序执行中止。

代码	错误信息	说明	对策
E4019	TCPIP) 插口(套接字)打开错误。 代码=XX DST. IP=XX. XX . XX. XX	当执行 TCP_LISTEN/TCP_CONNECT 命令时，未能打开套接字。已打开太多的套接字。或，参数错误如错误的 IP 地址。	此命令不要执行的太多。修改错误的参数，CODE(代码)表示 TCP/IP 包的错误子代码。
E4020	TCPIP) 插口(套接字)关闭错误。 代码=XX DST. IP=XX. XX . XX. XX	当执行 TCP_CLOSE/TCP_END_LISTEN 命令时，未能关闭套接字。可能是以太网上的问题或连接对方的问题。	检查连接对方的节点或连接线路是否有问题，如果需要的话，则进行修改。CODE(代码)表示 TCP/IP 包的错误子代码。
E4021	TCPIP) 通信错误。 代码=XX DST. IP=XX. XX . XX. XX	在执行 TCP_SEND/TCP_RECV 命令中，通信错误产生。可能是以太网上的问题或连接对方的问题。	检查连接对方的节点或连接线路是否有问题，如果需要的话，则进行修改。CODE(代码)表示 TCP/IP 包的错误子代码。
E4022	TCPIP) 消息(信息)太长。	当执行 TCP_SEND 命令时，试图发送的数据超过极限。	修改数据以便其数据量在极限范围内并重试程序。
E4023	TCPIP) 不能到达主机(上位机)。	当执行 TCP_CONNECT 命令时，不能与上位机连接。在网络上无该 IP 地址的上位机。端口号或 IP 地址可能错误。可能是以太网上的问题或连接对方的问题。	检查端口号或 IP 地址是否正确，如果必要的话，则进行修改。检查连接对方的节点或连接线路是否有问题，如果需要的话，则进行修改。
E4024	TCPIP) 通信时间超时。 DST. IP=XX. XX . XX. XX	在执行以 TCP 协议为基准的通信命令的过程中，超时产生。可能是以太网上的问题或连接对方的问题。	检查连接对方的节点或连接线路是否有问题，如果需要的话，则进行修改。
E4025	TCPIP) 连接失败。	当执行 TCP_CONNECT/TCP_SEND/TCP_RECV/TCP_ACCEPT 命令时，中断连接。可能是以太网上的问	检查连接对方的节点或连接线路是否有问题，如果需要的话，则进行修改。

代码	错误信息	说明	对策
		题或连接对方的问题。	
E4026	TCP/IP) 无缓冲空间。	在执行 TCP_SEND/TCP_RECV 命令的过程中，通信错误产生。TCP/IP 系统资源(队列，缓冲器)缺乏。或，可能是以太网上的问题或连接对方的问题。	同时，停止正在运行中的一些 TCP/IP 应用。重新启动控制器。
E4027	TCP/IP) 坏的插口。	当执行 TCP_CONNECT 命令时，不能连接。端口号或 IP 地址可能错误。可能是以太网上的问题或连接对方的问题。	检查端口号或 IP 地址是否正确，如果必要的话，则进行修改。检查连接对方的节点或连接线路是否有问题，如果需要的话，则进行修改。
E4056	TCP) 该端口不在 LISTEN(SOCK)。	当执行 TCP_ACCEPT 命令时，未能打开套接字。或当执行 TCP_END_LISTEN 命令时，未能关闭套接字。指定了用 TCP_LISTEN 命令未创建的套接字。	检查端口号是否正确，如果必要的话，则进行修改。
E4057	TCP) 套接字 ID 非法。	在执行 TCP_SEND/TCP_RECV 的过程中，通信错误产生。或，当执行 TCP_CLOSE/TCP_END_LISTEN 命令时，未能关闭套接字。可能已指定未创建的套接字号或端口号。或，程序的执行顺序错误。	检查套接字号或端口号是否正确，如果必要的话，则进行修改。检查程序的执行顺序是否错误，如果必要的话，则进行修正。
E4055	TCP) 不能创建套接字。	当执行 TCP_LISTEN/TCP_CONNECT 命令时，未能打开套接字。已打开太多的套接字。或，多次使用同一端口。	避免一次执行多个命令。修改参数。以前使用的端口用 TCP_CLOSE/TCP_END_LISTEN 命令关闭后，再使用。

6.2 套接字通信命令使用例

6.2.1 当机器人为服务器时

```
.PROGRAM main() ; 通信主程序
port = 49152
max_length = 255
tout_open = 60
tout_rec = 60
CALL open_socket ; 连接通信
IF sock_id < 0 THEN
    GOTO exit_end
END
text_id = 0
tout = 60
eret = 0
rret = 0
$data[1] = "001"
CALL send(eret,$data[1]) ; 指示处理 1
IF eret < 0 THEN
    PRINT "CODE 001 ERROR END code=",eret
    GOTO exit
END
CALL recv ; 接收处理 1 的结果
IF rret < 0 THEN
    PRINT "CODE 001 RECV ERROR END code=",rret
    GOTO exit
END
eret = 0
$data[1] = "002"
CALL send(eret,$data[1]) ; 指示处理 2
IF eret < 0 THEN
    PRINT "CODE 002 ERROR END code=",eret
    GOTO exit
END
CALL recv ; 接收处理 2 的结果
IF rret < 0 THEN
    PRINT "CODE 002 RECV ERROR END code=",rret
    GOTO exit
END
exit:
    CALL close_socket ; 断开通信
exit_end:
.END
```

```
.PROGRAM open_socket()      ; 开始通信
    er_count =0
listen:
    TCP_LISTEN ret1, port
    IF ret1<0 THEN
        IF er_count >= 5 THEN
            PRINT "不能与 PC 连接 (LISTEN)。停止程序。"
            sock_id = -1
            goto exit
        ELSE
            er_count = er_count+1
            PRINT "TCP_LISTEN error=",ret1," error count=",er_count
            GOTO listen
        END
    ELSE
        PRINT "TCP_LISTEN OK ",ret1
    END
    er_count =0
accept:
    TCP_ACCEPT sock_id, port, tout_open, ip[1]
    IF sock_id<0 THEN
        IF er_count >= 5 THEN
            PRINT "不能与 PC 连接 (ACCEPT)。停止程序。"
            TCP_END_LISTEN ret, port
            sock_id = -1
        ELSE
            er_count = er_count+1
            PRINT "TCP_ACCEPT error id=",sock_id," error count=",er_count
            GOTO accept
        END
    ELSE
        PRINT "TCP_ACCEPT OK id=",sock_id
    END
exit:
.END
.PROGRAM send(.ret,.data)           ; 通信 发送数据
    $send_buf[1] = .data
    buf_n = 1
    .ret = 1
    TCP_SEND sret, sock_id, $send_buf[1], buf_n, tout
    IF sret < 0 THEN
        .ret = -1
        PRINT "TCP_SEND error in SEND",sret
    ELSE
        PRINT "TCP_SEND OK  in SEND",sret
    END
.END
.PROGRAM recv()      ; 通信 接收数据
    .num=0
    TCP_RECV rret, sock_id, $recv_buf[1], .num, tout_rec, max_length
    IF rret < 0 THEN
        PRINT "TCP_RECV error in RECV",rret
        $recv_buf[1] = "000"
    ELSE
        IF .num > 0 THEN
            PRINT "TCP_RECV OK  in RECV",rret
        ELSE
            $recv_buf[1] = "000"
        END
    END
.END
```

```
.PROGRAM close_socket() ; 断开通信
    TCP_CLOSE ret,sock_id ; 通常套接字关闭
    IF ret < 0 THEN
        PRINT "TCP_CLOSE error ERROE=(",ret,") ",$ERROR(ret)
        TCP_CLOSE ret1,sock_id ; 套接字的强制关闭 (shutdown 关机)
        IF ret1 < 0 THEN
            PRINT "TCP_CLOSE error id=",sock_id
        END
    ELSE
        PRINT "TCP_CLOSE OK id=",sock_id
    END
    TCP_END_LISTEN ret,port
    IF ret < 0 THEN
        PRINT "TCP_CLOSE error id=",sock_id
    ELSE
        PRINT "TCP_CLOSE OK id=",sock_id
    END
.END
```

6.2.2 当机器人为客户端时

```
.PROGRAM main() ; 通信主程序
port = 49152
ip[1] = 192
ip[2] = 168
ip[3] = 0
ip[4] = 2
max_length = 255
tout_open = 60
tout_rec = 60
eret = 0
ret = 0
CALL open_socket ; 连接通信
IF sock_id < 0 THEN
    GOTO exit_end
END
text_id = 0
tout = 60
WHILE (1) DO
    ret = 0
    CALL recv ; 从 PC 接收指示数据
    IF ret < 0 THEN
        PRINT "Communication error code=", ret
        GOTO exit
    ELSE
        CASE ret OF
            VALUE 0:
                text_id = val($mid($recv_buf[1], 1, 3)) ; 读取指示号
            ANY:
        END
    END
    IF text_id > 0 THEN
        CALL com_test ; 各指示的处理
        IF eret < 0 THEN
            goto exit
        END
    END
END
exit:
    CALL close_socket ; 断开通信
exit_end:
.END
.PROGRAM com_test() ; 根据指示数据进行各处理
CASE text_id OF
    VALUE 0: ; 无指示
        GOTO break_exit
    VALUE 1: ; 指示处理 1
        ; 各处理
    VALUE 2: ; 指示处理 2
        ; 各处理
    VALUE 3: ; 指示处理 3
        ; 各处理
    any:
        GOTO break_exit
END
$end_buf[1] = $recv_buf[1] + " OK"
CALL send(eret, $end_buf[1]) ; 发送结束处理
break_exit:
.END
```

```
.PROGRAM open_socket()      ; 开始通信
    .er_count =0
connect:
    TCP_CONNECT sock_id, port, ip[1], tout_open
    IF sock_id<0 THEN
        IF .er_count >= 5 THEN
            PRINT "不能与 PC 连接。停止程序。"
        ELSE
            .er_count = .er_count+1
            PRINT "TCP_CONNECT error id=", sock_id, " error count=", .er_count
            GOTO connect
        END
    ELSE
        PRINT "TCP_CONNECT OK id=", sock_id
    END
.END
.PROGRAM recv()      ; 通信 接收数据
    .num=0
    TCP_RECV ret, sock_id, $recv_buf[1], .num, tout_rec, max_length
    IF ret < 0 THEN
        PRINT "TCP_RECV error in RECV", ret
        $recv_buf[1] = "000"
    ELSE
        IF .num > 0 THEN
            PRINT "TCP_RECV OK in RECV", ret
        ELSE
            $recv_buf[1] = "000"
        END
    END
.END
.PROGRAM send(.ret, .$data)      ; 通信 发送数据
    $send_buf[1] = .$data
    buf_n = 1
    .ret = 1
send_rt:
    TCP_SEND sret, sock_id, $send_buf[1], buf_n, tout
    IF sret < 0 THEN
        .ret = -1
        PRINT "TCP_SEND error in SEND", sret
    ELSE
        PRINT "TCP_SEND OK in SEND", sret
    END
.END
.PROGRAM close_socket()      ; 断开通信
    TCP_CLOSE ret, sock_id      ; 正常的套接字关闭
IF ret < 0 THEN
    PRINT "TCP_CLOSE error ERROE=(", ret, " ) ", $ERROR(ret)
    TCP_CLOSE ret1, sock_id      ; 套接字的强制关闭 (shutdown 关机)
    IF ret1 < 0 THEN
        PRINT "TCP_CLOSE error id=", sock_id
    END
ELSE
    PRINT "TCP_CLOSE OK id=", sock_id
END
.END
```

7.0 错误信息

代码	错误信息	说明	对策
E4013	TELNET) 发送错误。 代码=XX	当用 TCP/IP 发送数据时, 错误产生。可能是通讯线路有问题/中断或关闭连接。	检查连接对方的节点或连接线路是否有问题, 如果需要的话, 则进行修改。
E4014	TELNET) 接收错误。 代码=XX	当用 TCP/IP 接收数据时, 错误产生。可能是通讯线路有问题/中断或关闭连接。	检查连接对方的节点或连接线路是否有问题, 如果需要的话, 则进行修改。
E4015	TELNET) IAC(应用间通信)接收错误。代码=XX	当用 TCP/IP 接收数据时, 错误产生。可能是通讯线路有问题/中断或关闭连接。	检查连接对方的节点或连接线路是否有问题, 如果需要的话, 则进行修改。检查连接对方的 PC 的软件是否是 KCwinTCP/KRterm。
E4016	TELNET) 关闭失败。 代码=XX	用 TCP/IP 的 LISTEN/ACCEPT 功能时, 错误产生, 试图关闭套接字时, 也未成功。TCP/IP 系统资源(队列, 缓冲器)缺乏。或, 可能是通讯线路或连接对方的节点有问题。	同时, 停止正在运行中的一些 TCP/IP 应用。重新启动控制器。 CODE(代码)表示 TCP/IP 包的错误子代码。
E4017	TELNET) 主插口(套接字)关闭失败。代码=XX	用 TCP/IP 的 LISTEN/ACCEPT 功能时, 错误产生, 试图关闭主套接字时, 也未成功。TCP/IP 系统资源(队列, 缓冲器)缺乏。或, 可能是通讯线路或连接对方的节点有问题。	同时, 停止正在运行中的一些 TCP/IP 应用。重新启动控制器。 CODE(代码)表示 TCP/IP 包的错误子代码。
E4018	TELNET) 系统错误。 代码=XX	IPADDRESS 指令中的参数错误。	正确地设置 IPADDRESS 指令中的参数。 CODE(代码)表示 TCP/IP 包的错误子代码。
E4019	TCPIP) 插口(套接字)打开错误。代码=XX Dst. IP=XX. XX. X X. XX	当执行 UDP_SENDTO/UDP_RECVFROM/TCP_LISTEN/TCP_CONNECT 命令时, 未能打开套接字。已打开太多的套接字。或, 参数错误如错误的 IP 地址。	此命令不要执行的太多。修改错误的参数。 CODE(代码)表示 TCP/IP 包的错误子代码。

E4020	TCP/IP) 插口 (套接字) 关闭错误。代码 =XX Dst. IP=XX. XX. X X. XX	当执行 UDP_SENDTO/UDP_RECVFROM/TCP_CLOSE/TCP_END_LISTEN 命令时, 未能关闭套接字。可能是以太网或连接对方有问题。	检查连接对方的节点或连接线路是否有问题, 如果需要的话, 则进行修改。 CODE(代码)表示 TCP/IP 包的错误子代码。
E4021	TCP/IP) 通信错误。代码=XX Dst. IP=XX. XX. X X. XX	在执行 UDP_SENDTO/UDP_RECVFROM/TCP_SEND/TCP_RECV 命令中, 通信错误产生。可能是通讯线路有问题/中断或关闭连接。	检查连接对方的节点或连接线路是否有问题, 如果需要的话, 则进行修改。 CODE(代码)表示 TCP/IP 包的错误子代码。
E4022	TCP/IP) 消息 (信息) 太长。	当执行 UDP_SENDTO/UDP_RECVFROM/TCP_SEND 命令时, 试图发送的数据超过极限。	修改数据以便其数据量在极限范围内并重试程序。
E4023	TCP/IP) 不能到达主机 (上位机)。	当执行 UDP_SENDTO/UDP_RECVFROM/TCP_CONNECT 命令时, 不能与上位机连接。在网络上无该 IP 地址的上位机。端口号或 IP 地址可能错误。可能是通讯线路有问题/中断或关闭连接。	检查端口号或 IP 地址是否正确, 如果必要的话, 则进行修改。检查连接对方的节点或连接线路是否有问题, 如果需要的话, 则进行修改。
E4024	TCP/IP) 通信时间超时。 DST. IP=XX. XX. X X. XX	当执行以 TCP 协议为基准的通信命令时, 超时产生。可能是通讯线路有问题/中断或关闭连接。	检查连接对方的节点或连接线路是否有问题, 如果需要的话, 则进行修改。
E4025	TCP/IP) 连接失败。	当执行 UDP_SENDTO/UDP_RECVFROM/TCP_CONNECT/TCP_SEND/TCP_RECV/TCP_ACCEPT 命令时, 中断连接。可能是通讯线路有问题/中断或关闭连接。	检查连接对方的节点或连接线路是否有问题, 如果需要的话, 则进行修改。
E4026	TCP/IP) 无缓冲空间。	在执行 UDP_SENDTO/ UDP_RECVFROM/TCP_SEND/TCP_RECV 命令的过程中, 通信错误产生。TCP/IP 系统资源 (队列, 缓冲器) 缺乏。或, 可能是通讯线路有问题/中断或关闭连接。	同时, 停止正在运行中的一些 TCP/IP 应用。重新启动控制器。
E4027	TCP/IP) 坏的插口。	当执行 UDP_SENDTO/UDP_RECVFROM/TCP_CONNECT 命令时, 不能连接。端口号或 IP 地址可能错误。可能是以太网上的问题或连接对方的问题。	检查端口号或 IP 地址是否正确, 如果必要的话, 则进行修改。检查连接对方的节点或连接线路是否有问题, 如果需要的话, 则进行修改。

E4056	TCP) 该端口不在 LISTEN(SOCK)。	当执行 TCP_ACCEPT 命令时, 未能打开套接字。或当执行 TCP_END_LISTEN 命令时, 未能关闭套接字。指定了用 TCP_LISTEN 命令未创建的套接字。	检查端口号是否正确, 如果必要的话, 则进行修改。
E4057	TCP) 套接字 ID 非法。	在执行 TCP_SEND/TCP_RECV 命令的过程中, 通信错误产生。或, 当执行 TCP_CLOSE/TCP_END_LISTEN 命令时, 未能关闭套接字。可能已指定未创建的套接字号或端口号。或, 程序的执行顺序错误。	检查套接字号或端口号是否正确, 如果必要的话, 则进行修改。检查程序的执行顺序是否错误, 如果必要的话, 则进行修正。
E4055	TCP) 不能创建套接字。	当执行 TCP_LISTEN/TCP_CONNECT 命令时, 未能打开套接字。已打开太多的套接字。或, 多次使用同一端口。	避免一次执行多个命令。修改参数。以前使用的端口用 TCP_CLOSE/TCP_END_LISTEN 命令关闭后, 再使用。



川崎机器人控制器 E 系列
TCP/IP 通信功能手册

2012 年 1 月：第一版

川崎重工业株式会社出版

90210-1248DCA